# Solve equations that give {} result with complex quantity

I tried to find the roots of the following equation by using Solve:

Solve[8/(π u (-I a + γ) Sqrt[1 +
(h^2 v^2 u^2)/(-I a + γ)^2]) +
(16 (-I a + γ) (1 - Sqrt[1 +
(h^2 v^2 u^2)/(-I a + γ)^2]))/(h^2 π v^2 u^3) == 0, u]


but I did not get any result, it was only {}. I redefined (-I a + γ) as c based on some instructions I found here, but I got nothing.

• I tried Reduce in place of solve and it resulted in "False." – Fred Kline Oct 24 '16 at 6:06
• Trying to solve for an arbitrary x (to then take the limit x->0) solx = Reduce[expr == x, u]; returns a message ...A likely reason for this is that the solution set depends on branch \ cuts of Wolfram Language functions. However one can find a solution u==0 either by taking the limit or by looking at Series[expr, {u, 0, 2}]. – b.gates.you.know.what Oct 24 '16 at 6:18
• With; Reduce[expr == x, u], I got: u == -((2 (-2)^(3/4))/(Sqrt[h] Sqrt[[Pi]] Sqrt[v] Sqrt[x])) Takeing the limit of this quantity gives infinity! on the other hand when I use Series[expr, {u, 0, 2}],I got SeriesData[ u, 0, {Complex[0, 2] h^2 Pi^(-1) v^2 (a + Complex[0, 1] [Gamma])^(-3)}, 1, 3, 1]. How can I extract the exactly value of u? – Mariam Tohari Oct 24 '16 at 7:00
• Welcome to Mathematica.SE! 1) As you receive help, try to give it too, by answering questions in your area of expertise. 2) Take the tour and check the faqs! 3) When you see good questions and answers, vote them up by clicking the gray triangles, because the credibility of the system is based on the reputation gained by users sharing their knowledge. Also, please remember to accept the answer, if any, that solves your problem, by clicking the checkmark sign! – user9660 Oct 29 '16 at 4:51

A result of {} from Solve indicates that Solve thinks the solution set is empty. We can see that this is so, if we rationalize the left-hand side of the equation:

lhs = 8/(π u (-I a + γ) Sqrt[1 + (h^2 v^2 u^2)/(-I a + γ)^2]) +
(16 (-I a + γ) (1 - Sqrt[1 + (h^2 v^2 u^2)/(-I a + γ)^2]))/(h^2 π v^2 u^3);
conjugates = lhs /. {{e_ :> e}, {r : Sqrt[_] :> -r, r : 1/Sqrt[_] :> -r}};
Times @@ conjugates // Expand // Simplify
(* -(64/(π^2 u^2 (-a^2 + h^2 u^2 v^2 - 2 I a γ + γ^2)))  *)


Since the numerator is a constant 64, the equation lhs == 0 cannot be satisfied.

One might put the terms together and set the numerator equal to zero. But this yields a result that makes lhs undefined:

Solve[Numerator@Together@eqn == 0, u]
lhs /. %
(*
{{u -> 0}}
{Indeterminate}
*)


Whether or not this can be considered a valid solution depends on whether clearing the denominator can be justified in the problem in which the equation arose.

When you first do a few transformations by hand, you get a good result:

    8/(π u (-I a + γ) Sqrt[
1 + (h^2 u^2 v^2)/(-I a + γ)^2]) + (
16 (-I a + γ) (1 - Sqrt[
1 + (h^2 u^2 v^2)/(-I a + γ)^2]))/(h^2 π u^3 v^2) == 0


to

    8/(π u (-I a + γ) Sqrt[
1 + (h^2 u^2 v^2)/(-I a + γ)^2]) == -((
16 (-I a + γ) (1 - Sqrt[
1 + (h^2 u^2 v^2)/(-I a + γ)^2]))/(h^2 π u^3 v^2))


then

In[21]:= eqn =
8 h^2 π u^3 v^2 ==
FullSimplify[
ExpandAll[(-16 (-I a + γ) (1 - Sqrt[
1 + (h^2 u^2 v^2)/(-I a + γ)^2])) ( π u (-I a + \
γ) Sqrt[1 + (h^2 u^2 v^2)/(-I a + γ)^2])]]

Out[21]= 8 h^2 π u^3 v^2 ==
16 π u (h^2 u^2 v^2 + (a + I γ)^2 (-1 + Sqrt[
1 + (h^2 u^2 v^2)/(-I a + γ)^2]))

In[22]:= Solve[eqn, u]
Out[22]= {{u -> 0}}

In[23]:= Reduce[eqn, u]

Out[23]= (a h v + I h v γ != 0 && u == 0) || (u != 0 &&
a + I γ != 0 && v == 0) || (a + I γ != 0 && v == 0 &&
u == 0) || (a v + I v γ != 0 && h == 0 &&
u == 0) || (v != 0 && u != 0 && a + I γ != 0 && h == 0)

• I highly appreciate your cooperation! – Mariam Tohari Oct 24 '16 at 8:29
• but I wont a solution in which all the involved quantities must take any value except zero – Mariam Tohari Oct 24 '16 at 8:32
• This can only get fullfilled if a or [Gamma] or both together are +or - Infinity. Test this with Limit[ "nonzeropart of eqn",a->Infinity] – Akku14 Oct 24 '16 at 12:36